12.5 The Ant and the Blade of Grass

Stewart's Calculus: 6 ed. §3.1, 6ET ed. §2.7

Goal: In this project, you will use `D` to differentiate a function, construct a tangent line at a specific point and at a general point, plot a series of points connected by line segments, plot several things on the same graph, and solve an equation using `fsolve`.

Background: Before you start, try each of the following examples.

- Differentiating functions.
  If a function f is defined using an arrow definition, for example

  ```
  > f := x -> x^3;
  ```

  then its derivative is computed using Maple's `D` operator:

  ```
  > D(f);
  ```

  Notice that the output is an arrow-defined function but it doesn't have a name. If you wish to give it a name, say Df, then you type

  ```
  > Df:=D(f);
  ```

- Tangent lines.
  The line tangent to the curve $y = f(x)$ at the point $(a, f(a))$, is $y = f_{\tan}(x) = f(a) + f'(a)(x - a)$. For the function $f(x) = x^3$ discussed above, the tangent line at $x = 2$ can be defined as a function using:

  ```
  > a:=2;
  > ftan := x -> f(a) + Df(a)*(x-a);
  ```

  The equation of the tangent line is then

  ```
  > y = ftan(x);
  ```

  Hint: The benefit of defining `a:=2;` instead of just typing 2 each time, is that if you need to compute the tangent line at several points, you can cut and paste the whole sequence of commands and only change the value of `a` before re-executing.

- Plotting several functions.
  The function $y = f(x)$ and its tangent line $y = f_{\tan}(x)$, defined above, can be plotted together on the interval $-5 \le x \le 5$ by enclosing the two functions in square brackets `[ ]` and using the Maple `plot` command.

  ```
  > plot([f(x),ftan(x)], x=-5..5);
  ```

- Lists and plotting points.
  In Maple an ordered list is denoted by separating the items by commas and enclosing the list in square brackets `[ ]`. Thus, the point $(2, 5)$ is entered as `[2,5]`, while a list of points might be

  ```
  > octagon:=[ [1,0], [4,0], [5,1], [5,4], [4,5], [1,5],
  > [0,4], [0,1], [1,0] ];
  ```

If you plot this list of points and connect the dots, you should get an octagon.

```
> plot(octagon);
```

If you have several lists of points, enclose them in square brackets [ ]. Try the following.

```
> line1:=[ [1,3], [2,3] ]; line2:=[ [3,3], [4,3] ];
> you:=[ [1,2], [2,1], [3,1], [4,2] ];
> plot([octagon, line1, line2, you]);
```

What did you get?

Finally you can even plot lists of points together with functions and also specify the x-range, y-range and colors. For example

```
> plot([f(x), ftan(x), octagon, you], x=0..5, y=0..6,
> color=[red,blue,green,magenta]);
```

- Solving equations.
  Suppose you want to solve the equation $x/\pi + \sin(x) = 1$. You first enter the equation to make sure you have typed it properly.

  ```
  > eq := x/Pi + sin(x) = 1;
  ```

  You can then plot the left hand and right hand sides of this equation, to see where they intersect. Notice the use of the commands `lhs` and `rhs`.

  ```
  > plot([lhs(eq),rhs(eq)], x=-2*Pi..4*Pi);
  ```

  So there are three intersection points, i.e., solutions, one between 0 and 2, one between 2 and 4, and one between 4 and 6. To find the solutions, use Maple's `fsolve` command.

  ```
  > fsolve(eq,x);
  ```

  This finds the solution between 2 and 4. (It's π.) The other two can be found by adding ranges to `fsolve`:

  ```
  > fsolve(eq,x=0..2);   fsolve(eq,x=4..6);
  ```

  We now have all three solutions.

ASSIGNMENT: An ant is walking (to the right) over its ant mound, whose height (in cm) is given by the function: $h(x) = \dfrac{x^2 - 24x + 153928}{5(x^2 - 36x + 396)^2}$. Nearby there is a blade of grass, which is located as the line segment from $(40, 0.1)$ to $(40, 12)$. The goal in this lab is to find the point where the ant first sees the blade of grass. You can assume that the ant's line of sight is the tangent line to the ant mound. The following series of questions will lead you to the solution.

1. Define the function `h` that gives the height of the ant mound, using an arrow definition. Define the line segment occupied by the blade of grass and name it `grass`. Plot the ant mound in brown and the blade of grass in green on the same graph. Include the plot option, `scaling=constrained`.

2. Compute the derivative of `h` and name it `Dh`.

3. Compute the tangent line to $y = h(x)$ at $x = 16$ and define it as a function `htan` using an arrow definition. Plot the ant mound, the blade of grass, and the ant's line of sight when the ant is at $x = 16$. Can it see the blade of grass? Find the height H where the tangent line crosses the line $x = 40$ by evaluating `htan(40.)`.

4. Compute the tangent line to $y = h(x)$ at $x = 17.5$ and define it as a function `htan` using an arrow definition. Plot the ant mound, the blade of grass, and the ant's line of sight when the ant is at $x = 17.5$. Can it see the blade of grass? Find the height H where the tangent line crosses the line $x = 40$ by evaluating `htan(40.)`.

5. We can now see that, when the ant is at some position $x = a$ between 16 and 17.5, it can first see the top of the blade of grass. We want to find a. So, compute the tangent line to $y = h(x)$ at $x = a$ for a variable a. Define the tangent line at $x = a$ as a function `htan` using an arrow definition. NOTE: If you previously gave `a` a value, clear it by executing `a:='a';`

6. You can no longer plot the tangent line because its formula contains a variable, namely a. However, you can still find the height H where the tangent line crosses the line $x = 40$ by evaluating `htan(40.)`. When this height H equals the height of the blade of grass, the ant can just begin to see the blade of grass. Use Maple's `fsolve` command to solve for the value of a where H equals the height of the blade of grass. (You may need to specify a range for a in the `fsolve` command.) Denote the solution by `A`.

7. For the value A found in problem 6, compute the tangent line to $y = h(x)$ at $x = A$ and define it as a function `htan` using an arrow definition. Plot the ant mound, the blade of grass, and the ant's line of sight when the ant is at $x = A$. Can it see the blade of grass? Find the height H where the tangent line crosses the line $x = 40$ by evaluating `htan(40.)`.

8. There is a second solution to the equation $H = 12$. What is wrong with this solution?

9. *10% extra credit.* Animate the tangent line as the ant walks over the mound from $a = 7$ to $a = 19$. HINT: Use the `animate` command in the `plots` package.